

Άσκηση 733 - Πανελλαδικές 2014-2015 Επαναληπτικές, Θέμα 4°, Μονάδες 20

Μια πολυκατοικία έχει 5 ορόφους, με 8 διαμερίσματα ($\Delta 1, \Delta 2, \dots, \Delta 8$) σε κάθε όροφο. Τα διαμερίσματα $\Delta 1$ όλων των ορόφων έχουν το ίδιο εμβαδό ($E1$), τα διαμερίσματα $\Delta 2$ όλων των ορόφων έχουν το ίδιο εμβαδό ($E2$) κ.ο.κ. Το ποσό των κοινοχρήστων της πολυκατοικίας κατανέμεται στους 5 ορόφους, σύμφωνα με το ποσοστό συμμετοχής του κάθε ορόφου, όπως φαίνεται στον **Πίνακα III**.

Όροφος	Ποσοστό συμμετοχής
1 ^{ος}	5%
2 ^{ος}	15%
3 ^{ος}	20%
4 ^{ος}	25%
5 ^{ος}	35%

Το ποσό των κοινοχρήστων του κάθε ορόφου κατανέμεται στα διαμερίσματα του ορόφου αυτού, ανάλογα με το εμβαδό του καθενός διαμερίσματος.

Να γράψετε πρόγραμμα, το οποίο:

$\Delta 1$: Να περιλαμβάνει κατάλληλο τμήμα δηλώσεων. (**Μονάδες 2**)

$\Delta 2$: Να ζητάει:

i. Το συνολικό ποσό κοινοχρήστων της πολυκατοικίας (μονάδα 1).

ii. Τα εμβαδά $E1, E2, \dots, E8$ (μονάδα 1). (**Μονάδες 2**)

$\Delta 3$: Να υπολογίζει το ποσό των κοινοχρήστων που αναλογεί σε κάθε όροφο της πολυκατοικίας. (**Μονάδες 4**)

$\Delta 4$: Να υπολογίζει το ποσό των κοινοχρήστων που αναλογεί σε κάθε διαμέρισμα της πολυκατοικίας. (**Μονάδες 7**)

$\Delta 5$: Να αναζητά και να εμφανίζει τον αριθμό ορόφου (1-5) και τον αριθμό διαμερίσματος (1-8) ενός διαμερίσματος στο οποίο αναλογεί ποσό κοινοχρήστων μεγαλύτερο του μέσου όρου όλης της πολυκατοικίας. Η αναζήτηση να ξεκινά από τον 1^ο όροφο και για κάθε όροφο να ξεκινά από το διαμέρισμα $\Delta 8$. Η αναζήτηση να τερματίζεται μόλις βρεθεί ένα τέτοιο διαμέρισμα. (**Μονάδες 5**)

ΠΡΟΓΡΑΜΜΑ ΘΕΜΑ4_2015_Επ_ΑΣΚ_733

ΜΕΤΑΒΛΗΤΕΣ

ΑΚΕΡΑΙΕΣ: i, j

ΠΡΑΓΜΑΤΙΚΕΣ: Σύνολο, $EMB[8], OP[5], ΑθρΕμβ, ΔΙΑΜ[5, 8], Διαμέρισμα, Όροφος$

ΠΡΑΓΜΑΤΙΚΕΣ: sum, avg

ΛΟΓΙΚΕΣ: found

ΑΡΧΗ

! $\Delta 2$: Εισαγωγή συνολικού ποσού κοινοχρήστων & εμβαδών.

ΔΙΑΒΑΣΕ Σύνολο

ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 8

ΔΙΑΒΑΣΕ $EMB[j]$

ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

! $\Delta 3$: Υπολογισμός κοινοχρήστων ανά όροφο.

$OP[1] \leftarrow \text{Σύνολο} * 5/100$

$OP[2] \leftarrow \text{Σύνολο} * 15/100$

$OP[3] \leftarrow \text{Σύνολο} * 20/100$

$OP[4] \leftarrow \text{Σύνολο} * 25/100$

$OP[5] \leftarrow \text{Σύνολο} * 35/100$

```

! Δ4 : Υπολογισμός κοινοχρήστων ανά διαμέρισμα.
! Υπολογισμός συνολικού εμβαδού και των 8 διαμερισμάτων.
Αθρεμβ <-- 0
ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 8
    Αθρεμβ <-- Αθρεμβ + ΕΜΒ[j]
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

! Στο Αθρεμβ αντιστοιχεί ποσό ΟΡ[i]
! Στο ΕΜΒ[j] αντιστοιχεί ποσό x
! x = ΟΡ[i] * ΕΜΒ[j] / Αθρεμβ
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 5
    ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 8
        ΔΙΑΜ[i,j] <-- ΟΡ[i] * ΕΜΒ[j] / Αθρεμβ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

! Δ5 : Εύρεση διαμερίσματος με κοινόχρηστα μεγαλύτερα του μέσου όρου.
! Υπολογισμός μέσου όρου.
sum <-- 0
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 5
    ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 8
        sum <-- sum + ΔΙΑΜ[i,j]
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
avg <-- sum / 40

! Αναζήτηση στον πίνακα ΔΙΑΜ.Η αναζήτηση σταματά μόλις
! βρεθεί το πρώτο διαμέρισμα με κοινόχρηστα μεγαλύτερα του μέσου όρου.
found <-- ΨΕΥΔΗΣ
i <-- 1
ΟΣΟ (i <= 5) ΚΑΙ (found = ΨΕΥΔΗΣ) ΕΠΑΝΑΛΑΒΕ
    j <-- 8
    ΟΣΟ (j >= 1) ΚΑΙ (found = ΨΕΥΔΗΣ) ΕΠΑΝΑΛΑΒΕ
        ! Αν βρεθεί διαμέρισμα.
        ΑΝ ΔΙΑΜ[i,j] > avg ΤΟΤΕ
            ! Κρατάμε όροφο και διαμέρισμα και σταματάμε.
            Όροφος <-- i
            Διαμέρισμα <-- j
            found <-- ΑΛΗΘΗΣ
        ! Αλλιώς πάμε στο επόμενο διαμέρισμα.
        ΑΛΛΙΩΣ
            j <-- j - 1
        ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
    i <-- i + 1
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
! Αν βρέθηκε διαμέρισμα εμφανίζουμε όροφο και αριθμό.
ΑΝ found = ΑΛΗΘΗΣ ΤΟΤΕ
    ΓΡΑΨΕ 'Όροφος:', Όροφος
    ΓΡΑΨΕ 'Διαμέρισμα:', Διαμέρισμα
ΤΕΛΟΣ_ΑΝ

ΤΕΛΟΣ_ΠΡΟΓΡΑΜΜΑΤΟΣ

```